

13.7 Menü-Button

Die Entwicklungsumgebung (IDE) von Gambas benutzt mehrere *MenuButton* an unterschiedlichen Stellen in verschiedenen Symbolleisten:




Abbildung 13.7.1: MenuButton in der IDE von Gambas

Das macht Sinn, wenn man auf der Programmoberfläche auf kleinstem Platz ein vollständiges, jedoch disloziertes Menü verwenden will. Die Verwendung eines MenuButton als einzelner Button auf dem Formular ist ungewöhnlich. Man kann einen MenuButton nicht nur mit einem Menü ausrüsten, sondern ihm auch noch ein Kontextmenü zuordnen, wie Ihnen ein Blick in das Eigenschaftenfenster (F4) eines MenuButton zeigt.

13.7.1 Einsatz MenuButton

Wenn Sie einen oder mehrere MenuButton in einem Projekt auf einem Formular einsetzen wollen, sind folgende Hinweise zu beachten:

Zuerst sollten Sie die MenuButton in einem Container platzieren. Gut geeignet ist eine *ToolBar* für eine (horizontale) Symbolleiste:




Abbildung 13.7.1.1: Zwei MenuButton in einer Symbolleiste

Dann sind die einzelnen Menüs zu entwerfen und am sichersten mit dem MenüEditor in der gewünschten Struktur zu deklarieren:




Abbildung 13.7.1.2: Menü-Editor mit den einzelnen Menüs

Achten Sie darauf, die Eigenschaft *.Visible* für die Menüs in der obersten Hierarchie-Ebene (Bild-Datei und Hilfe) auf *False* zu setzen, damit diese nicht auf dem Formular erscheinen.

Für jedes Menü in der Menüliste – außer beim Menü *mnuSpace* – ist die Eigenschaft *.Caption* auf einen nicht leeren String zu setzen; auch wenn diese Zeichenkette nicht angezeigt wird, weil sonst ein Trennstrich erzeugt wird! Jedem Menü können Sie ein passendes Icon zuweisen.

Danach sind im Quelltext die Aktionen zu beschreiben, die ausgelöst werden, wenn ein bestimmtes Menü durch einen Klick aktiviert wird. Hier ein entsprechender Ausschnitt aus einem Quelltext:

```
Public Sub mnu11Open_Click()  
    OpenImage()  
    mnu13Print.Enabled = True  
End ' mnu11Open_Click()  
  
Public Sub mnu13Print_Click()  
    PrintImage()  
    mnu13Print.Enabled = False  
End ' mnu13Print_Click()
```

Abschließend sind u.a. die erwähnten Aktionen *OpenImage()* und *PrintImage()* zu codieren, welche im Wesentlichen die Funktionalität des Programms bestimmen:

```
' Eigene Programm-Aktionen  
  
Public Sub OpenImage()  
  
    Dialog.Title = "Importieren Sie eine Bild-Datei!"  
    Dialog.Filter = ["*.png", "Bild-Dateien"]  
  
    If Dialog.OpenFile(False) = True Then  
        Message.Info("Das Öffnen der Bild-Datei wurde abgebrochen!")  
        Return ' Cancel-Button gedrückt  
        ...  
    Endif ' Dialog.OpenFile  
  
End ' OpenImage()  
  
Public Sub PrintImage()  
    Message.Info("Das Bild wird gedruckt (Druck-Routine wird aufgerufen)!")  
End ' PrintImage()
```

13.7.2 Projekt MenuButton

Das vollständige Projekt finden Sie im Downloadbereich. Der Quelltext von *FMain.class* wird hier komplett angegeben:

```
[1] ' Gambas class file  
[2]  
[3] Public Sub Form_Open()  
[4]  
[5]     FMain.Center  
[6]     FMain.Resizable = False  
[7]     PictureBox1.Border = Border.None  
[8]     PictureBox1.Picture = Picture["fractal.jpg"]  
[9]     Label1.Text = "Fraktale Geometrie - Informatikkurs 2012"  
[10]    ToolBar1.H = 24  
[11]  
[12]    mnubtnFile.Width = 48  
[13]    mnubtnFile.H = 24  
[14]    mnubtnFile.Picture = Picture["icon:/16/options"]  
[15]    mnubtnFile.Menu = "mnuFile"  
[16]    mnubtnFile.Tooltip = "Datei-Operationen"  
[17]    mnubtnFile.MenuOnly = False  
[18]  
[19]    mnubtnHelp.Width = 72  
[20]    mnubtnHelp.H = 24  
[21]    mnubtnHelp.Text = "Hilfen"  
[22]    mnubtnHelp.Menu = "mnuHelp"  
[23]    mnubtnHelp.MenuOnly = True  
[24]  
[25]    toolbtnClose.Picture = Picture["icon:/16/quit"]  
[26]    toolbtnClose.Tooltip = "Programm beenden"  
[27]  
[28] End ' Form_Open()  
[29]  
[30] Public Sub Form_Close()  
[31]     FHelp.Close  
[32]     FMain.Close  
[33] End ' Form_Close()  
[34]
```

```

[35] ' Eigene Programm-Aktionen -----
[36]
[37] Public Sub OpenImage()
[38]
[39]     Dialog.Title = "Importieren Sie eine Bild-Datei!"
[40]     Dialog.Filter = ["*.png", "Bild-Dateien"]
[41]
[42]     If Dialog.OpenFile(False) = True Then ' Multiselect=False (Standard)
[43]         Message.Info("Das Öffnen der Bild-Datei wurde abgebrochen!")
[44]         Return ' Cancel-Button gedrückt
[45]     Else
[46]         Try PictureBox1.Picture = Picture.Load(Dialog.Path)
[47]         If Error Then
[48]             Message.Error("Fehler beim Öffnen der Bilddatei")
[49]         Else
[50]             Labell.Alignment = Align.Left
[51]             Labell.Text = " Datei: " & File.Name(Dialog.Path)
[52]         Endif ' ERROR?
[53]     Endif ' Dialog.OpenFile
[54]
[55] End ' OpenImage()
[56]
[57] Public Sub PrintImage()
[58]     Message.Info("Das Bild wird gedruckt (Druck-Routine)!")
[59] End ' PrintImage()
[60]
[61] ' Menü-Aktionen -----
[62]
[63] Public Sub mnullOpen_Click()
[64]     OpenImage()
[65]     mnul3Print.Enabled = True
[66] End ' mnullOpen_Click()
[67]
[68] Public Sub mnul3Print_Click()
[69]     PrintImage()
[70]     mnul3Print.Enabled = False
[71] End ' mnul3Print_Click()
[72]
[73] Public Sub mnu21Help_Click()
[74]     FHelp.Show
[75] End ' mnuHelp_Click()
[76]
[77] Public Sub mnu22Information_Click()
[78]     Desktop.Open("http://www.gambas-buch.de/dw/doku.php?id=k13:start")
[79] End ' mnu32Information_Click()
[80]
[81] Public Sub toolbtnClose_Click()
[82]     FMain.Close
[83] End ' toolbtnClose_Click()
[84]
[85] Public Sub mnubtnFile_Click()
[86]     Message.Info("Klick auf Symbol ----> Aufruf Unter-Programm oder Hilfetext ...\n
        Klick auf den kleinen schwarzen Pfeil ----> Ansicht Menüliste.")
[87] End ' mnubtnFile_Click()

```



Abbildung 13.7.2.1: MenuButton im Einsatz

Kommentare:

- Wenn Sie die Eigenschaft `.MenuOnly` eines `MenuButton` (Zeile 18) auf den Wert `False` setzen, dann können Sie einerseits mit einem Klick auf das Icon (1) ein Programm starten (Zeilen 85 bis 87), während andererseits ein Klick auf den kleinen schwarzen Pfeil (2) die Menüliste öffnet.

- Beim 2. MenuButton 'Hilfen' wird mit einem Klick auf den MenuButton – ohne Beachtung von Icon, Text oder dem Pfeil auf dem MenuButton – eine Menülite aufgerufen, mit der Sie eine lokale Hilfe (Alternative F1) oder eine globale Hilfe (Internet) aufrufen können, wie Sie der Abbildung 13.7.2.1 entnehmen können.
- Mit einem eleganten Klick auf den ToolButton (3) in der Symbolleiste beenden Sie das Programm.